
Abstract. The two flavours of DSL embedding are shallow and deep
embedding. In functional languages, shallow embedding models the lan-
guage constructs as functions in which the semantics are embedded.
Adding semantics is therefore cumbersome while adding constructs is
a breeze. Upgrading the functions to type classes lifts this limitation to
a certain extent.

Deeply embedded languages represent their language constructs as
data and the semantics are functions on it. As a result, the language
constructs are embedded in the semantics, hence adding new language
constructs is laborious where adding semantics is trouble free.

This paper shows that by abstracting the semantics functions in deep
embedding to type classes, it is possible to easily add language constructs
as well. So-called classy deep embedding results in DSLs that are exten-
sible both in language constructs and in semantics while maintaining a
concrete abstract syntax tree. Additionally, little type-level trickery or
complicated boilerplate code is required to achieve this.

Keywords: Functional programming · Haskell · Embedded
domain-specific languages

1 Introduction

The two flavours of DSL embedding are deep and shallow embedding [4]. In func-
tional programming languages, shallow embedding models language constructs
as functions in the host language. As a result, adding new language constructs—
extra functions—is easy. However, the semantics of the language is embedded
in these functions, making it troublesome to add semantics since it requires
updating all existing language constructs.

On the other hand, deep embedding models language constructs as data
in the host language. The semantics of the language are represented by func-
tions over the data. Consequently, adding new semantics, i.e. novel functions, is
straightforward. It can be stated that the language constructs are embedded in
the functions that form a semantics. If one wants to add a language construct,
all semantics functions must be revisited and revised to avoid ending up with
partial functions.

This juxtaposition has been known for many years [19] and discussed by many
others [11] but most famously dubbed the expression problem by Wadler [24]:

The expression problem is a new name for an old problem. The goal is to
define a data type by cases, where one can add new cases to the data type
and new functions over the data type, without recompiling existing code,
and while retaining static type safety (e.g., no casts).

In shallow embedding, abstracting the functions to type classes disentangles
the language constructs from the semantics, allowing extension both ways. This
technique is dubbed tagless-final embedding [5], nonetheless it is no silver bullet.
Some semantics that require an intensional analysis of the syntax tree, such as
transformation and optimisations, are difficult to implement in shallow embed-
ding due to the lack of an explicit data structure representing the abstract syntax
tree. The semantics of the DSL have to be combined and must hold some kind
of state or context, so that structural information is not lost [10].

1.1 Research Contribution

This paper shows how to apply the technique observed in tagless-final embedding
to deep embedding. The presented basic technique, christened classy deep embed-
ding, does not require advanced type system extensions to be used. However, it is
suitable for type system extensions such as generalised algebraic data types. While
this paper is written as a literate Haskell [18] program using some minor extensions
provided by GHC [23], the idea is applicable to other languages as well1.

2 Deep Embedding

Consider the simple language of integer literals and addition. In deep embedding,
terms in the language are represented by data in the host language. Hence,
defining the constructs is as simple as creating the following algebraic data type2.

data Expr0 = Lit0 Int
| Add0 Expr0 Expr0

Semantics are defined as functions on the Expr0 data type. For example, a
function transforming the term to an integer—an evaluator—is implemented as
follows.

eval0 :: Expr0 → Int
eval0 (Lit0 e) = e
eval0 (Add0 e1 e2) = eval0 e1 + eval0 e2

Adding semantics—e.g. a printer—just means adding another function while
the existing functions remain untouched. I.e. the key property of deep embedding.
1 Lubbers, M. (2022): Literate Haskell/lhs2TEX source code of the paper “Deep Embed-
ding with Class”: TFP 2022. Zenodo. https://doi.org/10.5281/zenodo.6650880.

2 All data types and functions are subscripted to indicate the evolution.

https://doi.org/10.5281/zenodo.6650880

The following function, transforming the Expr0 data type to a string, defines a
simple printer for our language.

print0 :: Expr0 → String
print0 (Lit0 v) = show v
print0 (Add0 e1 e2) = "(" ++ print0 e1 ++ "-" ++ print0 e2 ++ ")"

While the language is concise and elegant, it is not very expressive. Tradition-
ally, extending the language is achieved by adding a case to the Expr0 data type.
So, adding subtraction to the language results in the following revised data type.

data Expr0 = Lit0 Int
| Add0 Expr0 Expr0
| Sub0 Expr0 Expr0

Extending the DSL with language constructs exposes the Achilles’ heel of
deep embedding. Adding a case to the data type means that all semantics func-
tions have become partial and need to be updated to be able to handle this new
case. This does not seem like an insurmountable problem, but it does pose a
problem if either the functions or the data type itself are written by others or
are contained in a closed library.

3 Shallow Embedding

Conversely, let us see how this would be done in shallow embedding. First, the
data type is represented by functions in the host language with embedded seman-
tics. Therefore, the evaluators for literals and addition both become a function
in the host language as follows.

type Sems = Int
lits :: Int → Sems

lits i = i
adds :: Sems → Sems → Sems

adds e1 e2 = e1 + e2

Adding constructions to the language is done by adding functions. Hence,
the following function adds subtraction to our language.

subs :: Sems → Sems → Sems

subs e1 e2 = e1 − e2

Adding semantics on the other hand—e.g. a printer—is not that simple
because the semantics are part of the functions representing the language con-
structs. One way to add semantics is to change all functions to execute both
semantics at the same time. In our case this means changing the type of Sems

to be (Int ,String) so that all functions operate on a tuple containing the result
of the evaluator and the printed representation at the same time. Alternatively,
a single semantics can be defined that represents a fold over the language con-
structs [8], delaying the selection of semantics to the moment the fold is applied.

3.1 Tagless-Final Embedding

Tagless-final embedding overcomes the limitations of standard shallow embed-
ding. To upgrade to this embedding technique, the language constructs are
changed from functions to type classes. For our language this results in the
following type class definition.

class Exprt s where
litt :: Int → s
addt :: s → s → s

Semantics become data types3 implementing these type classes, resulting in
the following instance for the evaluator.

newtype Evalt = Et Int
instance Exprt Evalt where
litt v = Et v
addt (Et e1) (Et e2) = Et (e1 + e2)

Adding constructs—e.g. subtraction—just results in an extra type class and
corresponding instances.

class Subt s where
subt :: s → s → s

instance Subt Evalt where
subt (Et e1) (Et e2) = Et (e1 − e2)

Finally, adding semantics such as a printer over the language is achieved by
providing a data type representing the semantics accompanied by instances for
the language constructs.

newtype Printert = Pt String
instance Exprt Printert where
litt i = Pt (show i)
addt (Pt e1) (Pt e2) = Pt ("(" ++ e1 ++ "+" ++ e2 ++ ")")

instance Subt Printert where
subt (Pt e1) (Pt e2) = Pt ("(" ++ e1 ++ "-" ++ e2 ++ ")")

3 In this case newtypes are used instead of regular data declarations. A newtype is
a special data type with a single constructor containing a single value only to which
it is isomorphic. It allows the programmer to define separate class instances that
the instances of the isomorphic type without any overhead. During compilation the
constructor is completely removed [18, Sect. 4.2.3].

4 Lifting the Backends

Let us rethink the deeply embedded DSL design. Remember that in shallow
embedding, the semantics are embedded in the language construct functions.
Obtaining extensibility both in constructs and semantics was accomplished by
abstracting the semantics functions to type classes, making the constructs over-
loaded in the semantics. In deep embedding, the constructs are embedded in the
semantics functions instead. So, let us apply the same technique, i.e. make the
semantics overloaded in the language constructs by abstracting the semantics
functions to type classes. The same effect may be achieved when using simi-
lar techniques such as explicit dictionary passing or ML style modules. In our
language this results in the following type class.

class Eval1 v where
eval1 :: v → Int

data Expr1 = Lit1 Int
| Add1 Expr1 Expr1

Implementing the semantics type class instances for the Expr1 data type is
an elementary exercise. By a copy-paste and some modifications, we come to the
following implementation.

instance Eval1 Expr1 where
eval1 (Lit1 v) = v
eval1 (Add1 e1 e2) = eval1 e1 + eval1 e2

Subtraction can now be defined in a separate data type, leaving the original
data type intact. Instances for the additional semantics can now be implemented
separately as instances of the type classes.

data Sub1 = Sub1 Expr1 Expr1

instance Eval1 Sub1 where
eval1 (Sub1 e1 e2) = eval1 e1 − eval1 e2

5 Existential Data Types

The astute reader might have noticed that we have dissociated ourselves from the
original data type. It is only possible to create an expression with a subtraction
on the top level. The recursive knot is left untied and as a result, Sub1 can never
be reached from an Expr1.

Luckily, we can reconnect them by adding a special constructor to the Expr1
data type for housing extensions. It contains an existentially quantified [15] type
with type class constraints [13,14] for all semantics type classes [23, Sect. 6.4.6]
to allow it to house not just subtraction but any future extension.

data Expr2 = Lit2 Int
| Add2 Expr2 Expr2
| forall x .Eval2 x ⇒ Ext2 x

The implementation of the extension case in the semantics type classes is in
most cases just a matter of calling the function for the argument as can be seen
in the semantics instances shown below.

instance Eval2 Expr2 where
eval2 (Lit2 v) = v
eval2 (Add2 e1 e2) = eval2 e1 + eval2 e2
eval2 (Ext2 x ) = eval2 x

Adding language construct extensions in different data types does mean that
an extra Ext2 tag is introduced when using the extension. This burden can
be relieved by creating a smart constructor for it that automatically wraps the
extension with the Ext2 constructor so that it is of the type of the main data
type.

sub2 :: Expr2 → Expr2 → Expr2
sub2 e1 e2 = Ext2 (Sub2 e1 e2)

In our example this means that the programmer can write4:

e2 :: Expr2
e2 = Lit2 42 ‘sub2‘ Lit2 1

instead of having to write

e′
2 :: Expr2
e′
2 = Ext2 (Lit2 42 ‘Sub2‘ Lit2 1)

5.1 Unbraiding the Semantics from the Data

This approach does reveal a minor problem. Namely, that all semantics type
classes are braided into our datatypes via the Ext2 constructor. Say if we add
the printer again, the Ext2 constructor has to be modified to contain the printer
type class constraint as well5. Thus, if we add semantics, the main data type’s
type class constraints in the Ext2 constructor need to be updated. To avoid this,
the type classes can be bundled in a type class alias or type class collection as
follows.

class (Eval2 x , P rint2 x ) ⇒ Semantics2 x
data Expr2 = Lit2 Int

4 Backticks are used to use functions or constructors in an infix fashion [18, Sect. 4.3.3].
5 Resulting in the following constructor: forall x .(Eval2 x , P rint2 x ) ⇒ Ext2 x .

| Add2 Expr2 Expr2
| forall x .Semantics2 x ⇒ Ext2 x

The class alias removes the need for the programmer to visit the main data
type when adding additional semantics. Unfortunately, the compiler does need
to visit the main data type again. Some may argue that adding semantics hap-
pens less frequently than adding language constructs but in reality it means that
we have to concede that the language is not as easily extensible in semantics as
in language constructs. More exotic type system extensions such as constraint
kinds [3,25] can untangle the semantics from the data types by making the data
types parametrised by the particular semantics. However, by adding some boiler-
plate, even without this extension, the language constructs can be parametrised
by the semantics by putting the semantics functions in a data type. First the
data types for the language constructs are parametrised by the type variable d
as follows.

data Expr3 d = Lit3 Int
| Add3 (Expr3 d) (Expr3 d)
| forall x .Ext3 (d x ) x

data Sub3 d = Sub3 (Expr3 d) (Expr3 d)

The d type variable is inhabited by an explicit dictionary for the semantics,
i.e. a witness to the class instance. Therefore, for all semantics type classes, a
data type is made that contains the semantics function for the given semantics.
This means that for Eval3, a dictionary with the function EvalDict3 is defined,
a type class HasEval3 for retrieving the function from the dictionary and an
instance for HasEval3 for EvalDict3.

newtype EvalDict3 v = EvalDict3 (v → Int)
class HasEval3 d where
getEval3 :: d v → v → Int

instance HasEval3 EvalDict3 where
getEval3 (EvalDict3 e) = e

The instances for the type classes change as well according to the change in
the datatype. Given that there is a HasEval3 instance for the witness type d ,
we can provide an implementation of Eval3 for Expr3 d .

instance HasEval3 d ⇒ Eval3 (Expr3 d) where
eval3 (Lit3 v) = v
eval3 (Add3 e1 e2) = eval3 e1 + eval3 e2
eval3 (Ext3 d x ) = getEval3 d x

instance HasEval3 d ⇒ Eval3 (Sub3 d) where
eval3 (Sub3 e1 e2) = eval3 e1 − eval3 e2

Because the Ext3 constructor from Expr3 now contains a value of type d , the
smart constructor for Sub3 must somehow come up with this value. To achieve
this, a type class is introduced that allows the generation of such a dictionary.

class GDict a where
gdict :: a

This type class has individual instances for all semantics dictionaries, linking
the class instance to the witness value. I.e. if there is a type class instance known,
a witness value can be conjured using the gdict function.

instance Eval3 v ⇒ GDict (EvalDict3 v) where
gdict = EvalDict3 eval3

With these instances, the semantics function can be retrieved from the Ext3
constructor and in the smart constructors they can be generated as follows:

sub3 :: GDict (d (Sub3 d)) ⇒ Expr3 d → Expr3 d → Expr3 d
sub3 e1 e2 = Ext3 gdict (Sub3 e1 e2)

Finally, we reached the end goal, orthogonal extension of both language con-
structs as shown by adding subtraction to the language and in language seman-
tics. Adding the printer can now be done without touching the original code
as follows. First the printer type class, dictionaries and instances for GDict are
defined.

class Print3 v where
print3 :: v → String

newtype PrintDict3 v = PrintDict3 (v → String)
class HasPrint3 d where
getPrint3 :: d v → v → String

instance HasPrint3 PrintDict3 where
getPrint3 (PrintDict3 e) = e

instance Print3 v ⇒ GDict (PrintDict3 v) where
gdict = PrintDict3 print3

Then the instances for Print3 of all the language constructs can be defined.

instance HasPrint3 d ⇒ Print3 (Expr3 d) where
print3 (Lit3 v) = show v
print3 (Add3 e1 e2) = "(" ++ print3 e1 ++ "+" ++ print3 e2 ++ ")"
print3 (Ext3 d x ) = getPrint3 d x

instance HasPrint3 d ⇒ Print3 (Sub3 d) where
print3 (Sub3 e1 e2) = "(" ++ print3 e1 ++ "-" ++ print3 e2 ++ ")"

6 Transformation Semantics

Most semantics convert a term to some final representation and can be expressed
just by functions on the cases. However, the implementation of semantics such as
transformation or optimisation may benefit from a so-called intentional analysis
of the abstract syntax tree. In shallow embedding, the implementation for these
types of semantics is difficult because there is no tangible abstract syntax tree.
In off-the-shelf deep embedding this is effortless since the function can pattern
match on the constructor or structures of constructors.

To demonstrate intensional analyses in classy deep embedding we write an
optimizer that removes addition and subtraction by zero. In classy deep embed-
ding, adding new semantics means first adding a new type class housing the
function including the machinery for the extension constructor.

class Opt3 v where
opt3 :: v → v

newtype OptDict3 v = OptDict3 (v → v)
class HasOpt3 d where
getOpt3 :: d v → v → v

instance HasOpt3 OptDict3 where
getOpt3 (OptDict3 e) = e

instance Opt3 v ⇒ GDict (OptDict3 v) where
gdict = OptDict3 opt3

The implementation of the optimizer for the Expr3 data type is no compli-
cated task. The only interesting bit occurs in the Add3 constructor, where we
pattern match on the optimised children to determine whether an addition with
zero is performed. If this is the case, the addition is removed.

instance HasOpt3 d ⇒ Opt3 (Expr3 d) where
opt3 (Lit3 v) = Lit3 v
opt3 (Add3 e1 e2) = case (opt3 e1, opt3 e2) of

(Lit3 0, e′
2 ) → e′

2

(e′
1, Lit3 0) → e′

1

(e′
1, e′

2 ) → Add3 e′
1 e′

2

opt3 (Ext3 d x ) = Ext3 d (getOpt3 d x )

Replicating this for the Opt3 instance of Sub3 seems a clear-cut task at first
glance.

instance HasOpt3 d ⇒ Opt3 (Sub3 d) where
opt3 (Sub3 e1 e2) = case (opt3 e1, opt3 e2) of

(e′
1, Lit3 0) → e′

1

(e′
1, e

′
2 ) → Sub3 e′

1 e′
2

Unsurprisingly, this code is rejected by the compiler. When a literal zero is
matched as the right-hand side of a subtraction, the left-hand side of type Expr3

is returned. However, the type signature of the function dictates that it should
be of type Sub3. To overcome this problem we add a convolution constructor.

6.1 Convolution

Adding a loopback case or convolution constructor to Sub3 allows the removal of
the Sub3 constructor while remaining the Sub3 type. It should be noted that a
loopback case is only required if the transformation actually removes tags. This
changes the Sub3 data type as follows.

data Sub4 d = Sub4 (Expr4 d) (Expr4 d)
| SubLoop4 (Expr4 d)

instance HasEval4 d ⇒ Eval4 (Sub4 d) where
eval4 (Sub4 e1 e2) = eval4 e1 − eval4 e2
eval4 (SubLoop4 e1) = eval4 e1

With this loopback case in the toolbox, the following Sub instance optimises
away subtraction with zero literals.

instance HasOpt4 d ⇒ Opt4 (Sub4 d) where
opt4 (Sub4 e1 e2) = case (opt4 e1, opt4 e2) of

(e′
1, Lit4 0) → SubLoop4 e′

1

(e′
1, e

′
2 ) → Sub4 e′

1 e′
2

opt4 (SubLoop4 e) = SubLoop4 (opt4 e)

6.2 Pattern Matching

Pattern matching within datatypes and from an extension to the main data type
works out of the box. Cross-extensional pattern matching on the other hand—
matching on a particular extension—is something that requires a bit of extra care.
Take for example negation propagation and double negation elimination. Pattern
matching on values with an existential type is not possible without leveraging
dynamic typing [1,2]. To enable dynamic typing support, the Typeable type class
as provided by Data.Dynamic [22] is added to the list of constraints in all places
where we need to pattern match across extensions. As a result, the Typeable
type class constraints are added to the quantified type variable x of the Ext4
constructor and to ds in the smart constructors.

data Expr4 d = Lit4 Int
| Add4 (Expr4 d) (Expr4 d)
| forall x .Typeable x ⇒ Ext4 (d x ) x

First let us add negation to the language by defining a datatype represent-
ing it. Negation elimination requires the removal of negation constructors, so a
convolution constructor is defined as well.

data Neg4 d = Neg4 (Expr4 d)
| NegLoop4 (Expr4 d)

neg4 :: (Typeable d ,GDict (d (Neg4 d))) ⇒ Expr4 d → Expr4 d
neg4 e = Ext4 gdict (Neg4 e)

The evaluation and printer instances for the Neg4 datatype are defined as
follows.

instance HasEval4 d ⇒ Eval4 (Neg4 d) where
eval4 (Neg4 e) = negate (eval4 e)
eval4 (NegLoop4 e) = eval4 e

instance HasPrint4 d ⇒ Print4 (Neg4 d) where
print4 (Neg4 e) = "(~" ++ print4 e ++ ")"
print4 (NegLoop4 e) = print4 e

The Opt4 instance contains the interesting bit. If the sub expression of a
negation is an addition, negation is propagated downwards. If the sub expression
is again a negation, something that can only be found out by a dynamic pattern
match, it is replaced by a NegLoop4 constructor.

instance (Typeable d ,GDict (d (Neg4 d)),HasOpt4 d) ⇒
Opt4 (Neg4 d) where

opt4 (Neg4 (Add4 e1 e2))
= NegLoop4 (Add4 (opt4 (neg4 e1)) (opt4 (neg4 e2)))

opt4 (Neg4 (Ext4 d x ))
= case fromDynamic (toDyn (getOpt4 d x )) of
Just (Neg4 e) → NegLoop4 e

→ Neg4 (Ext4 d (getOpt4 d x ))
opt4 (Neg4 e) = Neg4 (opt4 e)
opt4 (NegLoop4 e) = NegLoop4 (opt4 e)

Loopback cases do make cross-extensional pattern matching less modular in
general. For example, Ext4 d (SubLoop4 (Lit4 0)) is equivalent to Lit4 0 in the
optimisation semantics and would require an extra pattern match. Fortunately,
this problem can be mitigated—if required—by just introducing an additional
optimisation semantics that removes loopback cases. Luckily, one does not need
to resort to these arguably blunt matters often. Dependent language functionality
often does not need to span extensions, i.e. it is possible to group them in the
same data type.

6.3 Chaining Semantics

Now that the data types are parametrised by the semantics a final problem needs
to be overcome. The data type is parametrised by the semantics, thus, using
multiple semantics, such as evaluation after optimising is not straightforwardly

possible. Luckily, a solution is readily at hand: introduce an ad-hoc combination
semantics.

data OptPrintDict4 v = OPD4 (OptDict4 v) (PrintDict4 v)
instance HasOpt4 OptPrintDict4 where
getOpt4 (OPD4 v ) = getOpt4 v

instance HasPrint4 OptPrintDict4 where
getPrint4 (OPD4 v) = getPrint4 v

instance (Opt4 v , P rint4 v) ⇒ GDict (OptPrintDict4 v) where
gdict = OPD4 gdict gdict

And this allows us to write print4 (opt4 e1) resulting in "((~42)+(~38))"
when e1 represents (∼ (42 + 38)) − 0 and is thus defined as follows.

e1 :: Expr4 OptPrintDict4
e1 = neg4 (Lit4 42 ‘Add4‘ Lit4 38) ‘sub4‘ Lit4 0

When using classy deep embedding to the fullest, the ability of the compiler to
infer very general types expires. As a consequence, defining reusable expressions
that are overloaded in their semantics requires quite some type class constraints
that cannot be inferred by the compiler (yet) if they use many extensions. Solving
this remains future work. For example, the expression ∼ (42 − 38) + 1 has to be
defined as:

e3 :: (Typeable d
,GDict (d (Neg4 d))
,GDict (d (Sub4 d))) ⇒ Expr4 d

e3 = neg4 (Lit4 42 ‘sub4‘ Lit4 38) ‘Add4‘ Lit4 1

7 Generalised Algebraic Data Types

Generalised algebraic data types (GADTs) are enriched data types that allow
the type instantiation of the constructor to be explicitly defined [7,9]. Lever-
aging GADTs, deeply embedded DSLs can be made statically type safe even
when different value types are supported. Even when GADTs are not supported
natively in the language, they can be simulated using embedding-projection pairs
or equivalence types [6, Sect. 2.2]. Where some solutions to the expression prob-
lem do not easily generalise to GADTs (see Sect. 9), classy deep embedding does.
Generalising the data structure of our DSL is fairly straightforward and to spice
things up a bit, we add an equality and boolean not language construct. To
make the existing DSL constructs more general, we relax the types of those con-
structors. For example, operations on integers now work on all numerals instead.
Moreover, the Litg constructor can be used to lift values of any type to the DSL
domain as long as they have a Show instance, required for the printer. Since some
optimisations on Notg remove constructors and therefore use cross-extensional

pattern matches, Typeable constraints are added to a. Furthermore, because the
optimisations for Addg and Subg are now more general, they do not only work for
Ints but for any type with a Num instance, the Eq constraint is added to these
constructors as well. Finally, not to repeat ourselves too much, we only show the
parts that substantially changed. The omitted definitions and implementation
can be found in Appendix A.

data Exprg d a where
Litg :: Show a ⇒ a → Exprg d a
Addg :: (Eq a,Num a) ⇒ Exprg d a → Exprg d a → Exprg d a
Extg :: Typeable x ⇒ d x → x a → Exprg d a

data Negg d a where
Negg :: (Typeable a,Num a) ⇒ Exprg d a → Negg d a
NegLoopg :: Exprg d a → Negg d a

data Notg d a where
Notg :: Exprg d Bool → Notg d Bool
NotLoopg :: Exprg d a → Notg d a

The smart constructors for the language extensions inherit the class con-
straints of their data types and include a Typeable constraint on the d type
variable for it to be usable in the Extg constructor as can be seen in the smart
constructor for Negg:

negg :: (Typeable d ,GDict (d (Negg d)),Typeable a,Num a) ⇒
Exprg d a → Exprg d a

negg e = Extg gdict (Negg e)
notg :: (Typeable d ,GDict (d (Notg d))) ⇒
Exprg d Bool → Exprg d Bool

notg e = Extg gdict (Notg e)

Upgrading the semantics type classes to support GADTs is done by an easy
textual search and replace. All occurrences of v are now parametrised by type
variable a:

class Evalg v where
evalg :: v a → a

class Printg v where
printg :: v a → String

class Optg v where
optg :: v a → v a

Now that the shape of the type classes has changed, the dictionary data types
and the type classes need to be adapted as well. The introduced type variable
a is not an argument to the type class, so it should not be an argument to the
dictionary data type. To represent this type class function, a rank-2 polymorphic
function is needed [23, Sect. 6.4.15] [17]. Concretely, for the evaluatior this results
in the following definitions:

newtype EvalDictg v = EvalDictg (forall a.v a → a)
class HasEvalg d where
getEvalg :: d v → v a → a

instance HasEvalg EvalDictg where
getEvalg (EvalDictg e) = e

The GDict type class is general enough, so the instances can remain the same.
The Evalg instance of GDict looks as follows:

instance Evalg v ⇒ GDict (EvalDictg v) where
gdict = EvalDictg evalg

Finally, the implementations for the instances can be ported without compli-
cation show using the optimisation instance of Notg:

instance (Typeable d ,GDict (d (Notg d)), HasOptg d) ⇒
Optg (Notg d) where

optg (Notg (Extg d x))
= case fromDynamic (toDyn (getOptg d x)) ::Maybe (Notg d Bool) of
Just (Notg e) → NotLoopg e

→ Notg (Extg d (getOptg d x))
optg (Notg e) = Notg (optg e)
optg (NotLoopg e) = NotLoopg (optg e)

8 Conclusion

Classy deep embedding is a novel organically grown embedding technique that
alleviates deep embedding from the extensibility problem in most cases.

By abstracting the semantics functions to type classes they become over-
loaded in the language constructs. Thus, making it possible to add new language
constructs in a separate type. These extensions are brought together in a spe-
cial extension constructor residing in the main data type. This extension case is
overloaded by the language construct using a data type containing the class dic-
tionary. As a result, orthogonal extension is possible for language constructs
and semantics using only little syntactic overhead or type annotations. The
basic technique only requires—well established through history and relatively
standard—existential data types. However, if needed, the technique generalises
to GADTs as well, adding rank-2 types to the list of type system requirements
as well. Finally, the abstract syntax tree remains observable which makes it suit-
able for intensional analyses, albeit using occasional dynamic typing for truly
cross-extensional transformations.

Defining reusable expressions overloaded in semantics or using multiple
semantics on a single expression requires some boilerplate still, getting around
this remains future work.

9 Related Work

Embedded DSL techniques in functional languages have been a topic of research
for many years, thus we do not claim a complete overview of related work.

Clearly, classy deep embedding bears most similarity to the Datatypes à la
Carte [21]. In Swierstra’s approach, semantics are lifted to type classes simi-
larly to classy deep embedding. Each language construct is their own datatype
parametrised by a type parameter. This parameter contains some type level rep-
resentation of language constructs that are in use. In classy deep embedding,
extensions do not have to be enumerated at the type level but are captured
in the extension case. Because all the constructs are expressed in the type sys-
tem, nifty type system tricks need to be employed to convince the compiler that
everything is type safe and the class constraints can be solved. Furthermore, it
requires some boilerplate code such as functor instances for the data types. In
return, pattern matching is easier and does not require dynamic typing. Classy
deep embedding only strains the programmer with writing the extension case
for the main data type and the occasional loopback constructor.

Löh and Hinze proposed a language extension that allows open data types
and open functions, i.e. functions and data types that can be extended with
more cases later on [12]. They hinted at the possibility of using type classes
for open functions but had serious concerns that pattern matching would be
crippled because constructors are becoming types, thus ultimately becoming
impossible to type. In contrast, this paper shows that pattern matching is eas-
ily attainable—albeit using dynamic types—and that the terms can be typed
without complicated type system extensions.

A technique similar to classy deep embedding was proposed by Najd and
Peyton Jones to tackle a slightly different problem, namely that of reusing a
data type for multiple purposes in a slightly different form [16]. For example
to decorate the abstract syntax tree of a compiler differently for each phase of
the compiler. They propose to add an extension descriptor as a type variable
to a data type and a type family that can be used to decorate constructors
with extra information and add additional constructors to the data type using
an extension constructor. Classy deep embedding works similarly but uses exis-
tentially quantified type variables to describe possible extensions instead of type
variables and type families. In classy deep embedding, the extensions do not need
to be encoded in the type system and less boilerplate is required. Furthermore,
pattern matching on extensions becomes a bit more complicated but in return it
allows for multiple extensions to be added orthogonally and avoids the necessity
for type system extensions.

Tagless-final embedding is the shallowly embedded counterpart of classy
deep embedding and was invented for the same purpose; overcoming the issues
with standard shallow embedding [5]. Classy deep embedding was organically
grown from observing the evolution of tagless-final embedding. The main dif-
ference between tagless-final embedding and classy deep embedding—and in
general between shallow and deep embedding—is that intensional analyses of

the abstract syntax tree is more difficult because there is no tangible abstract
syntax tree data structure. In classy deep embedding, it is possible to define
transformations even across extensions.

Hybrid approaches between deep and shallow embedding exist as well. For
example, Svenningson et al. show that by expressing the deeply embedded lan-
guage in a shallowly embedded core language, extensions can be made orthogo-
nally as well [20]. This paper differs from those approaches in the sense that it
does not require a core language in which all extensions need to be expressible.

Furthermore, I would like to thank Pieter and Rinus for the fruitful discussions, Ralf
for inspiring me to write a functional pearl, and the anonymous reviewers for their
valuable and honest comments.

A Appendix

A.1 Data Type Definitions

data Subg d a where
Subg :: (Eq a,Num a) ⇒ Exprg d a → Exprg d a → Subg d a
SubLoopg :: Exprg d a → Subg d a

data Eqg d a where
Eqg :: (Typeable a,Eq a) ⇒ Exprg d a → Exprg d a → Eqg d Bool
EqLoopg :: Exprg d a → Eqg d a

A.2 Smart Constructors

subg :: (Typeable d ,GDict (d (Subg d)),Eq a,Num a) ⇒
Exprg d a → Exprg d a → Exprg d a

subg e1 e2 = Extg gdict (Subg e1 e2)
eqg :: (Typeable d ,GDict (d (Eqg d)),Eq a,Typeable a) ⇒
Exprg d a → Exprg d a → Exprg d Bool

eqg e1 e2 = Extg gdict (Eqg e1 e2)

A.3 Semantics Classes and Data Types

newtype PrintDictg v = PrintDictg (forall a.v a → String)
class HasPrintg d where
getPrintg :: d v → v a → String

instance HasPrintg PrintDictg where
getPrintg (PrintDictg e) = e

newtype OptDictg v = OptDictg (forall a.v a → v a)
class HasOptg d where
getOptg :: d v → v a → v a

instance HasOptg OptDictg where
getOptg (OptDictg e) = e

A.4 GDict instances

instance Printg v ⇒ GDict (PrintDictg v) where
gdict = PrintDictg printg

instance Optg v ⇒ GDict (OptDictg v) where
gdict = OptDictg optg

A.5 Evaluator Instances

instance HasEvalg d ⇒ Evalg (Exprg d) where
evalg (Litg v) = v
evalg (Addg e1 e2) = evalg e1 + evalg e2
evalg (Extg d x ) = getEvalg d x

instance HasEvalg d ⇒ Evalg (Subg d) where
evalg (Subg e1 e2) = evalg e1 − evalg e2
evalg (SubLoopg e) = evalg e

instance HasEvalg d ⇒ Evalg (Negg d) where
evalg (Negg e) = negate (evalg e)
evalg (NegLoopg e) = evalg e

instance HasEvalg d ⇒ Evalg (Eqg d) where
evalg (Eqg e1 e2) = evalg e1 ≡ evalg e2
evalg (EqLoopg e) = evalg e

instance HasEvalg d ⇒ Evalg (Notg d) where
evalg (Notg e) = not (evalg e)
evalg (NotLoopg e) = evalg e

A.6 Printer Instances

instance HasPrintg d ⇒ Printg (Exprg d) where
printg (Litg v) = show v
printg (Addg e1 e2) = "(" ++ printg e1 ++ "+" ++ printg e2 ++ ")"
printg (Extg d x ) = getPrintg d x

instance HasPrintg d ⇒ Printg (Subg d) where
printg (Subg e1 e2) = "(" ++ printg e1 ++ "-" ++ printg e2 ++ ")"
printg (SubLoopg e) = printg e

instance HasPrintg d ⇒ Printg (Negg d) where
printg (Negg e) = "(negate " ++ printg e ++ ")"
printg (NegLoopg e) = printg e

instance HasPrintg d ⇒ Printg (Eqg d) where
printg (Eqg e1 e2) = "(" ++ printg e1 ++ "==" ++ printg e2 ++ ")"
printg (EqLoopg e) = printg e

instance HasPrintg d ⇒ Printg (Notg d) where
printg (Notg e) = "(not " ++ printg e ++ ")"
printg (NotLoopg e) = printg e

A.7 Optimisation Instances

instance HasOptg d ⇒ Optg (Exprg d) where
optg (Litg v) = Litg v
optg (Addg e1 e2) = case (optg e1, optg e2) of

(Litg 0, e′
2 ) → e′

2

(e′
1, Litg 0) → e′

1

(e′
1, e′

2 ) → Addg e′
1 e′

2

optg (Extg d x ) = Extg d (getOptg d x )

instance HasOptg d ⇒ Optg (Subg d) where
optg (Subg e1 e2) = case (optg e1, optg e2) of

(e′
1, Litg 0) → SubLoopg e′

1

(e′
1, e

′
2 ) → Subg e′

1 e′
2

optg (SubLoopg e) = SubLoopg (optg e)

instance (Typeable d ,GDict (d (Negg d)),HasOptg d) ⇒
Optg (Negg d) where

optg (Negg (Addg e1 e2))
= NegLoopg (Addg (optg (negg e1)) (optg (negg e2)))

optg (Negg (Extg d x ))
= case fromDynamic (toDyn (getOptg d x )) of
Just (Negg e) → NegLoopg e

→ Negg (Extg d (getOptg d x ))
optg (Negg e) = Negg (optg e)
optg (NegLoopg e) = NegLoopg (optg e)

instance HasOptg d ⇒ Optg (Eqg d) where
optg (Eqg e1 e2) = Eqg (optg e1) (optg e2)
optg (EqLoopg e) = EqLoopg (optg e)

17. Odersky, M., Läufer, K.: Putting type annotations to work. In: Proceedings of the
23rd ACM SIGPLAN-SIGACT Symposium on Principles of Programming Lan-
guages, pp. 54–67. POPL 1996, Association for Computing Machinery, New York
(1996). https://doi.org/10.1145/237721.237729

18. Peyton Jones, S. (ed.): Haskell 98 Language and Libraries: The Revised Report.
Cambridge University Press, Cambridge (2003)

19. Reynolds, J.C.: User-defined types and procedural data structures as complemen-
tary approaches to data abstraction. In: Gries, D. (ed.) Programming Methodology:
A Collection of Articles by Members of IFIP WG2.3, pp. 309–317. Springer, New
York (1978). https://doi.org/10.1007/978-1-4612-6315-9 22

20. Svenningsson, J., Axelsson, E.: Combining deep and shallow embedding for EDSL.
In: Loidl, H.-W., Peña, R. (eds.) TFP 2012. LNCS, vol. 7829, pp. 21–36. Springer,
Heidelberg (2013). https://doi.org/10.1007/978-3-642-40447-4 2

